

第3章 代数的データ型と型クラス

3.1 代数的データ型の定義

リストのようなデータ型をプログラマーがあらたに定義する方法も用意されている。このようなデータ型は複数の _____ (constructor) を持つことができる。そして個々の構成子はいくつかのフィールド (field) を持つことができる。このようなデータ型を _____ (algebraic datatype) という。

データ型の宣言の一般的な形式は次のとおりである。

```
data 型構成子名 型引数1 型引数2 ... 型引数k
  = 構成子名1 型1,1 ... 型1,n1
  | 構成子名2 型2,1 ... 型2,n2
  | ...
  | 構成子名m 型m,1 ... 型m,nm
```

型構成子名・構成子名ともに使える文字は変数名の場合と同じだが、変数名とは逆に _____ から始まる必要がある。この型は 構成子名₁ ~ 構成子名_m の m 種類のデータからなる。型_{i,1}, ..., 型_{i,n_i} は 構成子名_i が持つフィールドの型である。型引数₁, 型引数₂, ..., 型引数_k はフィールドの型の中に現れうる型変数である。ここで「|」は“または”と読む。例えば

```
1 data Foo x y = Alice x [y] | Bob String y | Charlie
```

というデータ型の場合、Foo Double Integer 型になるのは次のような式である、Alice 3.14 [1,2] あるいは Bob "Hello" 3 あるいは Charlie, ...

代数的データ型は構成子にフィールドがない場合は、C や Java の列挙 (enum) 型と同じようなものである。次の例では、

```
1 data Direction = North | South | West | East
2   deriving (Eq, Ord, Show)
```

North, South, West, East の 4 つが Direction 型を構成している。

(**deriving** ... については後述する。)

Q 3.1.1 じゃんけんの 3 つの手 (グー (Rock)・チョキ (Scissors)・パー (Paper)) を表すデータ型 Janken を定義せよ。(deriving ... はつけなくてよい。)

一般的には代数的データ型の構成子はフィールドを持つ。次の例は二分木 (binary tree) を表すデータ型を定義している。

```
1 data Tree a = Empty | Branch (Tree a) a (Tree a)
2                               deriving (Eq,Ord,Show)
```

このデータ型は Branch と Empty の 2 つの構成子を持つ。Empty はフィールドを持たず、それだけで二分木を構成する。Branch は 3 つのフィールドを持つ。1 番目と 3 番目は自分自身と同じ型の二分木であり、2 番目は要素である。つまり、Branch は次のような型を持っている。


```
Branch :: _____
```

ここで、a は _____ であり、ここには Integer や String などの具体的な型がはいることができる。例えば Tree Integer は要素が Integer 型であるような二分木の型である。Tree 自体は型ではなく型構成子 (type constructor) である。つまり、型パラメーターを伴ってはじめて型になる。

具体的には次のような式がこの Tree の型を持つ。

```
1 tree1 :: Tree a
2 tree1 = Empty
3 tree2 :: Tree Integer
4 tree2 = Branch Empty 1 Empty
5 tree3 :: Tree String
6 tree3 = Branch (Branch Empty "a" Empty)
7           "b" (Branch Empty "c" Empty)
8 tree4 :: Tree Integer
9 tree4 = Branch (Branch Empty 1 Empty)
10            2 (Branch Empty 3 (Branch Empty 4 Empty))
```

例えば tree4 の構造は、図で表すと次のようになる。



Tree に対する関数は、パターンマッチングを用いて、次のように定義することができる。つまり、仮引数の位置に構成子と変数からなる式を書くことができる。

```
1 top :: Tree a -> a
2 top (Branch _ a _) = a
3
4 isEmpty :: Tree a -> Bool
5 isEmpty Empty = True
```

```

6 isEmpty _      = False
7
8 -- 要素数、例えば treeSize tree4 は 4
9 treeSize :: Tree a -> Int
10 treeSize Empty      = 0
11 treeSize (Branch l a r) = treeSize l + 1 + treeSize r

```

つまり、top は二分木の先頭の要素を返す関数、isEmpty は空かどうかを判定する関数、treeSize は要素数を返す関数である。

Q 3.1.2 第 1 引数が第 2 引数に対して、Janken 型の勝ち・負け・引き分けのいずれであるかを判定する関数 judgeJanken :: Janken -> Janken -> Ordering を定義せよ。

ここで Ordering は次のように標準ライブラリーで定義された型である。

```

1 data Ordering = LT | EQ | GT
2               deriving (Eq, Ord, Bounded, Enum, Read, Show)

```

LT が負け、GT が勝ち、EQ が引き分け、を表すものとする

問 3.1.3 Tree 型に対して、次のような関数を定義せよ。

```

depth      :: Tree a -> Integer -- 深さ
preorder   :: Tree a -> [a]     -- 前順走査
inorder    :: Tree a -> [a]     -- 中順走査
postorder  :: Tree a -> [a]     -- 後順走査
reflect    :: Tree a -> Tree a  -- 鏡像

```

例えば、① depth tree4, ② preorder tree4, ③ inorder tree4, ④ postorder tree4, ⑤ reflect tree4 の結果はそれぞれ、① 3, ② [2, 1, 3, 4], ③ [1, 2, 3, 4], ④ [1, 4, 3, 2], ⑤ Branch (Branch (Branch Empty 4 Empty) 3 Empty) 2 (Branch Empty 1 Empty) となる。

問 3.1.4 一般の n 分木（任意の個数の部分木を持つことができる木、rose tree ともいう）を表すデータ型 RoseTree を定義せよ。また、n 分木の要素数を求める関数 roseSize :: RoseTree a -> Integer を定義せよ。

(発展) フィールドラベル

C の構造体のように代数的データ型の各フィールドに名前（フィールドラベル）をつけることも可能である。次のように、フィールド部分をブレース（{ ~ }）で囲み、「::」の前にフィールドラベルを書く。

```

1 data C = F { field1, field2 :: Int, field3 :: Bool }
2           deriving (Eq, Ord, Show)

```

この宣言は、次の宣言と同等のデータ型を定義する。

```
1 data C = F Int Int Bool deriving (Eq, Ord, Show)
```

さらに、フィールドラベルは新しいデータを構築するときにも、パターンマッチングにも使用することが出来る。いずれもコンストラクターの後に、ブレース内に「フィールドラベル = 式」のコンマ区切りの並びを書く。

```
1 v1 = F { field1 = 5, field2 = 12, field3 = False }
2
3 foo :: C -> Int
4 foo (F { field1 = x, field2 = y }) = x * x + y * y
```

このとき `foo v1` の値は $(5^2 + 12^2 =)$ 169 になる。

またフィールドラベルは、データからフィールドの値を取り出す関数として使用することが出来る。例えば、上で定義された `v1` に対して、`field1 v1` の値は 5 になる。

式の後に、「フィールドラベル = 式」のコンマ区切りの並びをブレース内に書いて、指定したフィールドの値のみを変更した新しいデータを構成するときにも使用できる。`v1 { field2 = 3 }` の値は `F { field1 = 5, field2 = 3, field3 = False }` になる。

3.2 型クラスとは

型クラスの説明に入る前にまずいくつかの用語を紹介する。

ポリモルフィズム（多相, polymorphism）とは関数（メソッド）などが _____ を許すことをいう。さらに、関数などがいろいろな型の引数を許し、しかも _____ ことを、_____（ad-hoc — その場限りの、という意味）多相という。オブジェクト指向言語の動的束縛（dynamic binding）はアドホック多相の一種である。

オブジェクト指向言語では、単にポリモルフィズムという言葉で動的束縛の意味まで含むことがある。

動的束縛と混同しがちな概念として _____（多重定義, overloading）がある。多重定義は一つの名前が複数の意味を持つことである。例えば `C` の「+」オペレーターは `int`（整数）型にも `double`（倍精度浮動小数点数）型にも適用できる。しかし最終的には、適用される型によって全く異なる機械語に翻訳される。動的束縛と異なる点は、多重定義はコンパイル（型チェック）時に解決されてしまう、という点である（つまり静的な束縛である）。実行時にオペランドの型に応じて処理を振り分けるようなことは行なわない。

つまり、多重定義もアドホック多相の実現方法の一つであると言えるが、適用範囲はコンパイル時に型がわかる場合に限定されてしまう。

Haskellのように型推論と多相型 (アドホックな多相に対して、型によって処理が変わらない多相のことをパラメトリック (parametric) な多相と言う。)を許す言語では、コンパイル時に得られる型情報では演算子の実装を決定することはできない。例えば、次のような関数:

```
twice x = x + x
```

を定義した時、 x の型が整数か浮動小数点数か決定できないので、「+」の実装も決定できない。つまり、CやJava (の演算子) で使われているような多重定義は、Haskellでは使えない。では、twiceはどのように実装され、どのような型を持つべきか?

一方、Haskellでは一つの代数的データ型の異なるコンストラクターのなかでは、パターンマッチングによりアドホック多相を実現している。例えば、Tree型のBranchとEmptyでは関数isEmptyやsizeの実装が異なる。上記のtwiceのような関数の定義を可能にするためには複数の型にまたがるアドホック多相を実現する必要がある。そこで、Haskellでは という仕組みが導入されている。

3.3 Haskell の型クラス

以下では Haskell の型クラス (type class) を説明する。例として、Eq という型クラスを取り上げる。

HaskellでもCと同じように「==」 (等号) オペレーターはInteger (整数) 型にもDouble (倍精度浮動小数点数) 型にも、さらに他のいくつかの型にも適用できる。一方、関数型は比較できない。例えば $(\lambda x \rightarrow x + x) == (\lambda x \rightarrow 2 * x)$ はエラーである。Haskellは多相を許す言語であるので、例えば、

```
1 member x []      = False
2 member x (y:ys) = x == y || member x ys
3
4 subset xs ys     = all (\ x -> member x ys) xs
```

という関数memberやsubsetを定義すると、member 5 [1, 4, 7] のように [Integer] (整数のリスト) 型の引数にも、member "Kagawa" ["Tokushima", "Ehime", "Kochi"] のように [String] (文字列のリスト) 型の引数にも適用できる。しかし、member $(\lambda x \rightarrow x + x)$ $(\lambda x \rightarrow x + 1, \lambda x \rightarrow 2 * x)$ のように関数のリストに適用することはできない。

ここでmemberやsubsetの型は、単なる $a \rightarrow [a] \rightarrow \text{Bool}$ や $[a] \rightarrow [a] \rightarrow \text{Bool}$ ではない。それでは、memberやsubsetはどのような型を持ち、どのように実装されているのであろうか? SchemeやPythonのように動的に型付けされる言語では、各データオブジェクトが型の情報をつねに保持しているので、実行時に型に応じて適切な関数を選択することができる。しかし、Haskellのようにコンパイル時に型チェックを行なう言語では、実行時にはデータは型の情報を保持していないのが普通である。

Haskell では、これらの関数の型は自動的に推論される（型推論 — ただし、その方法の詳細については本稿で取り扱う範囲を超える）。型推論の結果だけを示すと、`member`と`subset`は次のような型を持っている。

```
member :: _____
subset :: _____
```

ここで“`Eq a =>`”という部分は、`a`という型変数が`Eq`という型の集まり（_____）に属していなければいけない、という型に関する制約（type constraint）を表す。`Eq`という型クラスは、「`==`」（等号）が定義されているような型の集まりのことである。一般に型クラスとは、_____のことである。

型クラスは通常のオブジェクト指向言語でのクラス・インスタンスという言葉とは意味が異なるので注意する。通常のオブジェクト指向言語ではクラスは_____（つまり型）であるのに対し、Haskell の型クラスは_____である。（Java のインタフェースの概念に似ている（というよりも、順序から言えば Java のインタフェースが Haskell の型クラスに似ている、というほうが適切である。）。）

3.4 クラス宣言とインスタンス宣言

型クラスの定義には `class` というキーワードを用いる。例えば、`Eq` クラスの定義は Haskell では次のように書く。（Prelude に定義済み。）

```
1 class Eq a where
2   (==), (/=) :: a -> a -> Bool   -- != ではないので注意
3   a /= b = not (a == b)        -- デフォルトの定義
```

これが、「型 `a` が型クラス `Eq` に属するためには、`a -> a -> Bool` という型を持つ 2 つの関数 `(==)`, `(/=)` を持たなければいけない」という意味になる。一般的には、

```
class クラス0 型変数0 where
  関数1 :: 型1
  関数2 :: 型2
  ...
```

のようにキーワード `class` のあとにクラス名と型変数を書き、キーワード `where` の後に関数の型宣言を並べる。これは、型変数₀ がクラス₀ に属するためには、型₁ という型をもつ関数₁、型₂ という型をもつ関数₂ ... が必要である（関数でなくても良いのだが、説明を簡単にするため、すべて関数ということにした。）、という意味である。

そして例えば以前紹介した `Direction` 型が（`deriving` 節がなかったとして）`Eq` クラスに属する（`Direction` が `Eq` の_____である）ことを宣言するためには、`instance` というキーワードを用いる。

```
1 instance Eq Direction where
2   North == North = True
3   South == South = True
4   West  == West  = True
5   East  == East  = True
6   _     == _     = False
```

ここで「(/=)」演算子については、クラス宣言の中でデフォルトの定義が用意されているので、インスタンス宣言では定義を省略することができる。

また、Tree 型の場合 (deriving 節がなかったとして)、次のように宣言する。

```
1 instance Eq a => Eq (Tree a) where
2   Empty      == Empty      = True
3   Branch l1 n1 r1 == Branch l2 n2 r2
4               = l1 == l2 && n1 == n2 && r1 ==
5   r2
6   _          == _          = False
```

“Eq a =>” の部分は Tree a に等号を定義するためには、要素の型である a に等号が定義されていなければいけないという制約を表す。Tree のようにパラメーターを持つ型の場合、こういう制約が必要な場合が多い。

一般的には、

```
instance 制約0 => クラス0 型0 where
  関数1 = 式1
  関数2 = 式2
  ...
```

のようにキーワード instance のあとに制約とクラス名、型を書き、where というキーワードのあとに関数の定義を並べる。これは、制約₀ のもとで、型₀ は、クラス₀ のインスタンスであり、関数₁、関数₂ の実装はそれぞれ式₁、式₂ である、という意味である。制約は、一般に (クラス₁ 型変数₁, クラス₂ 型変数₂, ...) という形式である。

ほとんどの型 (例えばリストや組) は Eq クラスに属するが、関数型については、一般に 2 つの関数が等価であるかどうかを判定することは原理的に不可能なので、Eq クラスに属さない。

Q 3.4.1 組込みの Bool 型と等価なデータ型:

```
data MyBool = MyTrue | MyFalse
```

を Eq クラスのインスタンスとして宣言せよ。(なお Bool 型に対する Eq クラスのインスタンスは標準ライブラリー中で宣言されているので、宣言をプログラム中に書く必要はない。)

Q 3.4.2 以前の Quiz で定義したじゃんけん (Janken) 型を Eq クラスのインスタンスとして宣言せよ。例えば `Rock == Rock` は `True`、`Scissors == Paper` は `False` である。

Q 3.4.3 “`size :: a -> Int`” というメソッドを持つクラス `Sizable` クラスを定義せよ。例えば、

```
1 instance Sizable [a] where
2     size xs = length xs
3
4 instance Sizable (Tree a) where
5     size t = treeSize t
```

のようにインスタンスを宣言すると、`size [1,5,2]` は 3 に、`size tree4` は 4 になる。

3.5 その他の標準的な型クラス

`Eq` の他に実用上重要な型クラスとして、`Ord`, `Show`, `Num` などがある。(以下の説明用コードでは主要でないメソッドは省略している。)

```
1 class Eq a => Ord a where      -- Ord は order (順序) の
   こと
2     (<), (<=), (>=), (>) :: a -> a -> Bool
3     max, min                :: a -> a -> a
4     -- ...
5
6 class Show a where
7     show                    :: a -> String
8     -- ...
9
10 class (Eq a, Show a) => Num a where
11     (+), (-), (*)          :: a -> a -> a
12     fromInteger           :: Integer -> a
13     -- ...
14
15 class Num a => Fractional a where
16     (/)                   :: a -> a -> a
17     -- ...
```


Ord は不等号、Show は文字列への変換、Num と Fractional は四則演算のメソッドを定義している型クラスである。

クラス宣言の「=>」の左側にあるクラスは、スーパークラスと呼ばれる。

例えば、Eq は Ord のスーパークラスである。これは、例えば Ord クラスのインスタンスになる型は、必ず Eq クラスのインスタンスでもなければならない、ということの意味する。

少し余談になるが、型クラス宣言の文法を決めるときにスーパークラスを示す矢印は逆向きに（つまり `class Eq a <= Ord a where ...` のように）するべきだった、という議論がある。これは `Ord τ` という制約が、`Eq τ` という制約を含意しているためである。恐らく、そのように定めるほうがすっきりしていたのだろうが、今さら一度決まった文法は変えられない。

実は、これまで触れていなかったが、`1` や `3.14` などの数値リテラルは、Haskell ではそれぞれ、

```
1    :: Num t => t
3.14 :: Fractional t => t
```

という型を持っている。だから、例えば `1` という数値リテラルは、`Int` としても、`Double` としても使用できる。

Show, Eq, Ord クラスなどに対するインスタンス宣言は、ほとんどのデータ型で必要になり、しかも同じような定義になるので、データ型の宣言が `deriving` /di'raivɪŋ/ というキーワードを持っていれば、Haskell の処理系がこれらのインスタンス宣言を自動的に生成してくれることになっている。例えば `Tree` については次のように書く。

```
1 data Tree a = Empty | Branch (Tree a) a (Tree a)
2                deriving (Eq, Ord, Show)
```

これで、`Tree` に対して期待どおりの `(==)`, `(>)`, `show` などのメソッドが定義される。

Q 3.5.1 組込みのリスト型と等価なデータ型

```
1 data MyList a = MyNil | MyCons a (MyList a)
```

を、`deriving` を用いずに、`Eq` クラスと `Ord` クラスのインスタンスとして宣言せよ。`Ord` クラスのメソッドにはいわゆる辞書式の順序を用いよ。

(クラスの定義中にデフォルトの実装が定義されているので、`Eq` クラスの `==` メソッドと `Ord` クラスの `<=` メソッドだけを定義すれば、他のメソッドの定義は自動的に生成される。)

ヒント: 次のような、リストから MyList への変換を行う関数を定義して、

```
1 toMyList :: [a] -> MyList a
2 toMyList [] = MyNil
3 toMyList (x:xs) = MyCons x (toMyList xs)
```

いくつかのケースをテストせよ。

```
1 toMyList "abc" <= toMyList "abd" -- True
2 toMyList "ab" <= toMyList "abc" -- True
3 toMyList "ab" <= toMyList "a" -- False
4 toMyList "ab" <= toMyList "ba" -- True
```

3.6 オブジェクト指向との関係

オブジェクト指向言語で使用される概念との関連について触れる。

オブジェクト指向を特徴づけるキーワードとして動的束縛 (dynamic binding) ・
_____ (inheritance) ・ _____ (encapsulation) の3つがよく挙げられる。

Q 3.6.1 左の語句の意味と対応する右の説明を線で結べ。

- | | | |
|-------|---|----------------------------------|
| 動的束縛 | ・ | クラスの実装の詳細を他のクラスから隠すことができること |
| 継承 | ・ | 呼び出されるメソッドの実装がオブジェクトの実行時の型で決まること |
| カプセル化 | ・ | 上位クラスのメソッドの実装を下位クラスで再利用できること |

カプセル化と継承は、ポリモルフィズムと動的束縛があっこそ意味がある概念である。ポリモルフィズムがあるから、実装の詳細を入れ替えても再コンパイルせずにコードを実行することができる。また、動的束縛があるから、継承したメソッドの意味がクラスに応じて自動的に変わり、同じようなメソッドを何度も定義する必要がなくなる。そのため、プログラミング言語の実装という観点から見れば、ポリモルフィズム、特に動的束縛こそがオブジェクト指向の本質であると言っても良い。

3.7 オブジェクト指向のクラスと代数的データ型

Haskell や ML といった関数型言語では、複数の構成子 (constructor) からなる代数的データ型 (algebraic data type) を定義できる。代数的データ型を構成する各構成子を、オブジェクト指向型言語のクラスに相当すると見なせば、関数型言語もアドホック多相や動的束縛に相当する機構を持っている。関数は、さまざま

まな構成子の引数を受け取り、また呼び出されるコードがパターンマッチングにより、オブジェクトの実行時の構成子で定まる。例えば、Tree 型の isEmpty という関数は Branch と Empty で実行されるコードが変わる。

オブジェクト指向言語のクラスと関数型言語の代数的データ型の構成子の違いは、拡張性の方向である。代数的データ型は、既存の構成子にそれを引数とする新しい関数を追加していくのは可能だが、既存の関数に新しい構成子を追加することはできない。（例えば組み込みの代数的データ型であるリスト型に [] と (:) 以外の新しい構成子を後から追加することはできない。）逆にクラスは既存の関数（メソッド）を新しいデータ型（クラス）に定義することは可能だが、既存のデータ型（クラス）に新しい関数（メソッド）を追加することはできない。（例えば、Java の組み込みのクラスである String クラスに、新しいメソッドを後から追加することはできない。静的 (static) なメソッドなら追加できるが、動的束縛を伴わせることはできない。Kotlin の拡張関数も動的束縛ではない。）

型クラスは、関数型言語にオブジェクト指向言語と同じ方向の拡張性（つまり既存の関数に新しい型を引数として追加する）を付与する仕組み、あるいはオブジェクト指向言語の動的束縛を関数型言語で解釈する仕組みと考えることもできる。

ただし、オブジェクト指向言語は新しいデータ型（クラス）を定義するときには、新しい関数（メソッド）を追加することはできない。これに対して型クラスは既存の型に新しい関数（メソッド）を定義することができる。オブジェクト指向言語でこれに同等なことを考えれば、既存のクラスに新しい（Java の）インターフェースの実装を追加することに相当する。

3.8 型クラスとサブタイピング

オブジェクト指向言語では、あるスーパークラスを継承するクラスに属するオブジェクトは、一つの変数に代入したり、一つのコンテナにまとめたりすることができる。しかし、Haskell では、例えば Integer と Char と [Integer] はすべて Show クラスに属するが、それらを一つのリストにまとめることはできない。つまり、次のような複数の型を持つ要素を含むリストは型付けできない。

```
1 -- 実際には部分式の [1, 'a', [1, 4, 7]] が型付け不可なので
2 -- 全体の impossible も型付け不可
3 impossible :: [String]    -- とはいかない
4 impossible = map show [1, 'a', [1, 4, 7]]
```

しかし、Show の場合、String に変換した結果がわかれば良いのだから、次のようなリストなら構成することができる。

```
1 possible :: [String]
2 possible = [show 1, show 'a', show [1, 4, 7]]
```

もう少し一般的な例として、次のような型クラスを考える。

```

1 class C a where
2   foo :: a -> Int -> Bool
3   bar :: a -> a
4   baz :: a -> Char -> a

```

これに対して、次のような型と変換関数を考えることができる。

```

1 data T = T (Int -> Bool)  -- foo の型に対応
2                 T        -- bar の型に対応
3                 (Char -> T)  -- baz の型に対応
4
5 instance C T where
6   foo (T f g h) n = f n
7   bar (T f g h)   = g
8   baz (T f g h) c = h c
9
10 toT :: C a => a -> T
11 toT a = T (foo a) (toT (bar a)) (\ c -> toT (baz a c))

```

構成子 T の 3 つの構成要素の型、 $\text{Int} \rightarrow \text{Bool}$ と T と $\text{Char} \rightarrow T$ は、それぞれ $\text{foo} :: C\ a \Rightarrow a \rightarrow \underline{\text{Int} \rightarrow \text{Bool}}$ と $\text{bar} :: C\ a \Rightarrow a \rightarrow \underline{a}$ と $\text{baz} :: C\ a \Rightarrow a \rightarrow \underline{\text{Char} \rightarrow a}$ に対応している。下線部の型の中の型変数 a を自身の型 T に置き換える。

このとき、 a が C のインスタンスの型を持つならば、 a と $\text{toT}\ a :: T$ は、 $\text{foo}, \text{bar}, \text{baz}$ に対して、どれも同じように振る舞う。さらに、 a, b, c がすべて C のインスタンスだが、異なる型に属していたとしても、 $\text{toT}\ a, \text{toT}\ b, \text{toT}\ c$ はどれも T 型であり、一つのリストにまとめることができる。

一般的に型クラス

```

class C  $\alpha$  where
   $m_1 :: \alpha \rightarrow \tau_1$ 
   $m_2 :: \alpha \rightarrow \tau_2$ 
   $m_3 :: \alpha \rightarrow \tau_3$ 

```

の各メソッドの戻り値の型 τ_1, τ_2, τ_3 に、対象の型変数 α が引数の型の位置（つまり \rightarrow の左側に）に現れないのならば—戻り値の型の位置に現れるのは構わない—変換先の型の T 型:

```

data T = T  $\sigma_1 \sigma_2 \sigma_3$ 

```

（ただし σ_i は τ_i 中の α の出現を T で置き換えた型である。）と $\text{toT} :: C\ \alpha \Rightarrow \alpha \rightarrow T$ のような変換関数を定義することができる。変換したオブジェクトは元のオブジェクトと同じ振る舞いをし、単一の型なので一つのコンテナにまとめることができる。この手法については、参考文献 (Odersky 1991) に詳しい説明がある。

3.9 (参考) 型クラスの問題点

型推論とアドホック多相をうまく両立した型クラスだが、いくつかの問題点も残っている。

わかりにくいエラーメッセージ

型クラスはエラーメッセージがわかりにくい、またはエラーになってほしいものがそもそもエラーにならない、という欠点がある。参考文献 (Heeren & Hage 2005) にいくつか例が挙げられている。

曖昧さ (ambiguity)

例えば、次のようなクラス定義があるとする。これは実際の Show, Read クラスを少し単純化している。

```
1 class Show a where
2   show :: a -> String
3
4 class Read a where
5   read :: String -> a
```

この定義の元で、次のような関数を定義する。

```
1 foo x = show (read x)
```

この関数の型は、`foo :: (Show a, Read a) => String -> String` となる。型変数 `a` の型は、`=>` の左辺にしか現れないので、型推論で決定できない。(すなわち曖昧である。) すると次のプログラムのように、

```
1 foo x = show (read x :: Integer)
```

`a` の具体的な型をユーザが明示しなければ意味が定まらなくなってしまう。(プログラムの意味が型宣言に依存することになり、気持ちが悪い。)

3.10 (参考) Dictionary-Passing Style 変換

ここからは、Haskell が型クラスをどのように実装しているかを説明する。(ただし、このような実装方法が Haskell の仕様で定められているわけではない。あくまでも良く使われる実装方法の一例である。)

クラス宣言・インスタンス宣言や制約された型 (`... => ...`) を持つ関数は、コンパイル時にそれらを用いない普通の関数やデータの定義に書き換えられる。

まず、クラス宣言はメソッドを要素に持つような型の宣言とアクセサの定義に翻訳される。例えば Eq クラスの場合、

```
1 type Eq' a = (a -> a -> Bool, a -> a -> Bool)
2
3 eq' :: Eq' a -> (a -> a -> Bool)
4 eq' = \ (e, _) -> e          -- (==) に対応する
5
```

```
6 ne' :: Eq' a -> (a -> a -> Bool)
7 ne' = \ (_, n) -> n          -- (/=) に対応する
```

この Eq' a 型のようなオブジェクトは一般に _____ (method dictionary) と呼ばれる。

インスタンス宣言は具体的な型を持つメソッド辞書の定義に翻訳される。例えば、instance Eq Direction は次のように Eq' Direction 型のオブジェクトの定義になる。

```
1 eqDirectionDic :: Eq' Direction
2 eqDirectionDic = (eqDirection,
3                  \ a b -> not (a `eqDirection` b))
4   where North `eqDirection` North = True
5         South `eqDirection` South = True
6         West `eqDirection` West = True
7         East `eqDirection` East = True
8         _ `eqDirection` _ = False
9
10 eqTreeDic :: Eq' a -> Eq' (Tree a)
11 eqTreeDic (eqA, _) = (eqTree, \ a b -> not (eqTree a
12                                     b))
13   where Empty `eqTree` Empty = True
14         Branch l1 n1 r1 `eqTree` Branch l2 n2 r2
15           = l1 `eqTree` l2 && n1 `eqA` n2 && r1 `eqTree`
16           r2
17         _ `eqTree` _ = False
```

そして型クラスを使っている (... => ... という型を持つ) 関数の定義は、コンパイル時に次のようにメソッド辞書 (Eq' a 型) を追加の引数とする関数の定義に書き換えられている。メソッド辞書には、通常は関数が含まれるので、これらの関数は高階関数になる。

```
1 member' :: Eq' a -> a -> [a] -> Bool
2 member' d x [] = False
3 member' d x (y:ys) = eq' d x y || member' d x ys
4
5 subset' :: Eq' a -> [a] -> [a] -> Bool
6 subset' d xs ys = all (\ x -> member' d x ys) xs
```

これらの関数の呼出しは次のように型に応じて具体的なメソッド辞書を渡される形に書き換えられる。

```
member North [North, South, West]
  ↳ member' _____ North [North, South, West]
subset [North, South] [North, South, West]
  ↳ subset' _____ [North, South]
                                     [North, South, West]
```

このような書き換え (Dictionary-Passing Style 変換) は Haskell ではコンパイル中の型推論時に自動的に行なわれる。つまり、アドホック多相の実行時のコストは、辞書オブジェクトの中から関数を取り出し、それを起動するだけにな

る。要するに、アドホック多相は高階関数で解釈できる(ただし高階関数になるので、最適化が難しくなってしまう可能性はある。)。動的に(つまり実行時に)メソッドを探しているように見えて、実際には静的に(コンパイル時に)ほとんどの必要な処理が済んでいる。

問 3.10.1 次のように定義されている関数 `lookup`:

```
1 data Maybe a = Just a | Nothing
2
3 lookup :: Eq a => a -> [(a, b)] -> Maybe b
4 lookup x ((n,v):rest)
5     = if n == x then Just v else lookup x rest
6 lookup x [] = Nothing
```

(`lookup` は標準ライブラリーに定義済みの関数である。) の Dictionary-Passing Style 変換後の形 `lookup'`:

```
lookup' :: Eq' a -> a -> [(a, b)] -> Maybe b
```

を示せ。例えば、`lookup 1 [(1,2), (4,7)]` と `lookup' eqIntDic 1 [(1,2), (4,7)]` の値が、あるいは `lookup 1.1 [(1.4,0.1), (4.2,6.9)]` と `lookup' eqDoubleDic 1.1 [(1.4,0.1), (4.2,6.9)]` の値が同じ値になる。(`eqIntDic`, `eqDoubleDic` は各自で適宜定義する。)

```
1 lookup' :: Eq' a -> a -> [(a, b)] -> Maybe b
2 lookup' d x ((n,v):rest)
3     = if eq' d n x then Just v else lookup' d x rest
4 lookup' d x [] = Nothing
```

3.11 (参考) 他のオブジェクト指向言語について

一般的なオブジェクト指向言語でも、たいていは“メソッド辞書”に対応するデータを扱うことで、動的束縛を実現していると考えられる。つまり、“隠れた”高階関数である。しかし、関数の独立した引数としてではなく、オブジェクトに付随する形になっていることが多いと思われる。つまり、各オブジェクトがクラスに対応するデータ構造へのポインターを持っていて、クラスに対応するデータ構造がメソッドの辞書を含んでいるという場合が多い。Smalltalk のメソッド呼出しの実装の方法は、例えば参考文献 (Guzdial & Rose 2003) の第 6 章に説明されている。

一方、代数的データ型の場合は、メソッド辞書に相当するものは各関数に付随しているかたちになる。

JavaScript は、クラスではなく (prototype) という概念に基づくオブジェクト指向を採用している。(ちなみにプロトタイプ方式を最初に広めた言語は Self という言語である。) JavaScript のメソッド呼出しの仕組みは参考文献 (久野 2001) の第 6 章に解説がある。ただし、最近の JavaScript はクラスも採り入れている。

Common Lispのオブジェクト指向拡張 (CLOS) は多重メソッド (multi-method) と言って、他の多くのオブジェクト指向言語と異なり、2つ以上のパラメーターの型 (クラス) によって実際に呼出すメソッドの実装を決定する仕組みを持っている。

問 3.11.1 実際のオブジェクト指向言語 (Smalltalk (Guzdial & Rose 2003), CLOS, JavaScript (久野 2001), C++, Java (Lindholm & Yellin 2001), Python, Ruby など) で動的束縛や継承がどのように実装されているか調べよ。

3.12 まとめ

型クラスは、Haskell でアドホック多相を可能にするための仕組みである。既存の関数を新しいデータ型に拡張するための仕組みでもある。コンパイル時に高階関数への変換が行なわれるので、実行時のコストはほとんどかからない。しかし、わかりにくいエラーメッセージなど、いくつかの問題は残っている。

3.13 さらに詳しく知りたい人のために...

文献 (Wadler & Blott 1988) は、型クラスのアイディアを最初に紹介した論文である。文献 (Hall et al. 1996) は、現在の Haskell の型クラスを詳しく説明している。文献 (Jones 2000) は、Haskell の (型クラスに関する部分を含む) 型推論を、具体的に Haskell のプログラムを用いて説明している。文献 (Odersky 1991) は、オブジェクト指向言語のサブタイプと同様の効果を、関数型言語の枠組みで得るための手法を提案している。文献 (Heeren & Hage 2005) は、型クラスのエラーメッセージの問題点と改良法について詳しく述べている。

(Wadler & Blott 1988) Philip Wadler and Stephen Blott, "How to make *ad-hoc* polymorphism less *ad-hoc*" Conference Record of the Sixteenth Annual ACM Symposium on Principles of Programming Languages, pp. 60–73, 1988 年 10 月

(Hall et al. 1996) Cordelia Hall, Kevin Hammond, Simon Peyton Jones and Philip Wadler, "Type Classes in Haskell" ACM Transactions on Programming Languages and Systems 18 巻 2 号, pp. 109–138, 1996 年

(Jones 2000) Mark P. Jones, "Typing Haskell in Haskell"
<https://web.cecs.pdx.edu/~mpj/thih/>, 2000 年 11 月

(Odersky 1991) Martin Odersky, "Objects and Subtyping in a Functional Perspective" IBM Research Report RC 16423, 1991 年 1 月

(Heeren & Hage 2005) Bastiaan Heeren and Jurriaan Hage, "Type Class Directives" Seventh International Symposium on Practical Aspects of Declarative Languages (LNCS 3350), pp.253–267, 2005 年 1 月

(Guzdial & Rose 2003) Mark Guzdial and Kim Rose 編、軌音組訳
「Squeak 入門 過去から来た未来のプログラミング環境」2003 年 3 月 星雲社

(久野 2001) 久野 靖 「入門 JavaScript」2001 年 8 月 ASCII

(Lindholm & Yellin 2001) Tim Lindholm and Frank Yellin 著、村上 雅
章 訳「Java 仮想マシン仕様 第2版」 2001 年 5 月 ピアソン・エデュケー
ション
